Pensieve header: The main ${\mathfrak g}_0$ theorem.

Reminders

1. Fill out “Graduate Course Participation Form” (see http://drorbn.net/AcademicPensieve/Classes/17-1350-AKT/GCP.jpg, in time).
2. Make sure that you have Mathematica and that you play with these programs!
3. Change meeting time?



$R=\sum a_i\otimes b_i \in A\otimes A = \widehat{U(\mathfrak g)}\otimes\widehat{U(\mathfrak g)}$

s.t. $R^{12}R^{13}R^{23}=R^{23}R^{13}R^{12}$

$\to \sum_{i,j,k} b_i a_j b_k a_i b_j a_k \in U(\mathfrak g)$



Today: $\mathfrak g_0 = \langle h,e,l,f\rangle / h \text{ central},\ [e,l]=-e \quad [f,l]=f \quad [e,f]=h$

$r=h\otimes l+e\otimes f \qquad R=\exp(r)$

Note $U(\mathfrak g_0)^{\otimes S} = U(\oplus_S \mathfrak g_0) = U(\langle h_i,e_i,l_i,f_i\rangle / h_i \text{ central},\ [e_i,l_j]=\delta_{ij}e_i \text{ etc.})$

Theorem. $R=e^{h\otimes l+e\otimes f}=\mathbb{O}\left(\exp\left(hl+\frac{e^h-1}{h}ef\right)\,\middle|\, e\otimes lf\right)$.

Representing g_0

```
Date: Sun, 23 May 1999 17:24:53 +0200 (MET DST)
From: Dylan Thurston <Dylan.Thurston@math.unige.ch>
To: Dror Bar-Natan <drorbn@math.huji.ac.il>
Subject: Oh, yeah...

Duflo had another possibly useful comment: he said that there is a simpler (non-trivial) algebra than
sl_2, given by the relations (more or less)

        [x,y] = z
        [x,t] = x
        [y,t] = -y
        [t,z] = [x,z] = [y,z] = 0

I hope I got that right; it should be the Lie algebra of matrices

        0 x z
        0 t y
        0 0 0

He says that one can write down the exponential explicitly, that commutators very quickly go to zero,
etc.  Apparently it plays the same role for reducible lie algebras that sl_2 plays for semi-simple ones.
It’s worth thinking about.

Best,
        Dylan
```

$$\rho h = \begin{pmatrix} 0 & 0 & -1 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}; \rho e = \begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{pmatrix}; \rho l = \begin{pmatrix} 0 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 0 \end{pmatrix}; \rho f = \begin{pmatrix} 0 & 1 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix};$$

```
B[x_?MatrixQ, y_?MatrixQ] := x.y - y.x;
```

```
MatrixForm /@ {B[ρe, ρl], B[ρf, ρl], B[ρe, ρf], B[ρh, ρe], B[ρh, ρl], B[ρh, ρf]}
```

$$\left\{\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & -1 \\ 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 1 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & -1 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}\right\}$$

Implementing g_0

```
PBWRule = {e → 1, l → 2, f → 3};
B[U@e, U@l] = -U@e;
B[U@f, U@l] = U@f;
B[U@e, U@f] = h U[];
```

```
$TD = 3; ħ /: ħ^d_. /; d > $TD := 0;
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. PBWRule]; x_ ≺ y_ := ! OrderedQ[{y, x} /. PBWRule];
Simp[ℰ_] := Collect[ℰ, _U, Expand];
```

```
U_i_[ℰ_] := ℰ /. {h → h_i, t → t_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
B[U[(x_)_i_], U[(y_)_i_]] := B[U[x_i], U[y_i]] = U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, x_] = 0;
B[U[y_], U[x_]] := B[U[y], U[x]] = Simp[-B[U[x], U[y]]];
B[x_, y_] := x ** y - y ** x;
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

```
U[xx___, x_] ** U[y_, yy___] := If[x ≤ y, U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
```

```
UU[l___, x_^n_., r___] := UU[l, Sequence @@ Table[x, {n}], r];
UU[l___, 1, r___] := UU[l, r];
UU[] = U[];
UU[l_, r___] := U[l] ** UU[r];
```

```
UProducts[{}, 0] = {UU[]};
UProducts[{}, n_Integer] /; n > 0 = {};
UProducts[{x_, xs___}, n_Integer] :=
  Sort@Flatten@Table[UU[x^k] ** u, {k, 0, n}, {u, UProducts[{xs}, n - k]}];
UProducts[xs_List, k_Integer, n_Integer] := UProducts[Flatten@Table[x_j, {x, xs}, {j, k}], n];
UProducts[any__, {n_}] := Flatten@Table[UProducts[any, k], {k, 0, n}];
```

```
r_{i_,j_} := Simp[ħ (h_i UU[l_j] + UU[e_i, f_j])]
```

```
UExp[u_] := Module[{s, t, k},
   s = t = U[]; k = 0;
   While[k < 20 ∧ 0 =!= (t = t ** u), s += t / (++k) !];
   Simp[s]];
R_{i_,j_} := UExp[r_{i,j}];
```

```
m[i_, j_, k_][ℰ_] := Simp[ℰ /. {
     u_U :> UU @@ Join[DeleteCases[u, x_{i|j}], U @@ Cases[u, x_i_ :> x_k], U @@ Cases[u, x_j_ :> x_k]],
     h_{i|j} → h_k}]
```

The Invariant of the Trefoil

 ; $TD = 2; R_{4,1} ** R_{2,5} ** R_{6,3}

$$U[] + \hbar h_4 U[l_1] + \hbar h_6 U[l_3] + \hbar h_2 U[l_5] + \left(\hbar + \frac{\hbar^2 h_2}{2}\right) U[e_2, f_5] + \left(\hbar + \frac{\hbar^2 h_4}{2}\right) U[e_4, f_1] + \left(\hbar + \frac{\hbar^2 h_6}{2}\right) U[e_6, f_3] + \frac{1}{2}\hbar^2 h_4^2 U[l_1, l_1] + \hbar^2 h_4 h_6 U[l_1, l_3] + \hbar^2 h_2 h_4 U[l_1, l_5] + \frac{1}{2}\hbar^2 h_6^2 U[l_3, l_3] + \hbar^2 h_2 h_6 U[l_3, l_5] + \frac{1}{2}\hbar^2 h_2^2 U[l_5, l_5] + \hbar^2 h_4 U[e_2, l_1, f_5] + \hbar^2 h_6 U[e_2, l_3, f_5] + \hbar^2 h_2 U[e_2, l_5, f_5] + \hbar^2 h_4 U[e_4, l_1, f_1] + \hbar^2 h_6 U[e_4, l_3, f_1] + \hbar^2 h_2 U[e_4, l_5, f_1] + \hbar^2 h_4 U[e_6, l_1, f_3] + \hbar^2 h_6 U[e_6, l_3, f_3] + \hbar^2 h_2 U[e_6, l_5, f_3] + \frac{1}{2}\hbar^2 U[e_2, e_2, f_5, f_5] + \hbar^2 U[e_2, e_4, f_1, f_5] + \hbar^2 U[e_2, e_6, f_3, f_5] + \frac{1}{2}\hbar^2 U[e_4, e_4, f_1, f_1] + \hbar^2 U[e_4, e_6, f_1, f_3] + \frac{1}{2}\hbar^2 U[e_6, e_6, f_3, f_3]$$

```
Timing[$TD = 3; R_{4,1} ** R_{2,5} ** R_{6,3} // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] // m[1, 5, 1] // m[1, 6, 1]]
```

$$\left\{0.625,\ \left(1 - 2\hbar h_1 + \hbar^2 h_1^2 + \frac{2}{3}\hbar^3 h_1^3\right) U[] + \left(3\hbar h_1 - 6\hbar^2 h_1^2 + 3\hbar^3 h_1^3\right) U[l_1] + \left(3\hbar - \frac{3\hbar^2 h_1}{2} - \frac{3}{2}\hbar^3 h_1^2\right) U[e_1, f_1] + \left(\frac{9}{2}\hbar^2 h_1^2 - 9\hbar^3 h_1^3\right) U[l_1, l_1] + \left(9\hbar^2 h_1 - \frac{9}{2}\hbar^3 h_1^2\right) U[e_1, l_1, f_1] + \frac{9}{2}\hbar^3 h_1^3 U[l_1, l_1, l_1] + \left(\frac{9\hbar^2}{2} + \frac{9\hbar^3 h_1}{2}\right) U[e_1, e_1, f_1, f_1] + \frac{27}{2}\hbar^3 h_1^2 U[e_1, l_1, l_1, f_1] + \frac{27}{2}\hbar^3 h_1 U[e_1, e_1, l_1, f_1, f_1] + \frac{9}{2}\hbar^3 U[e_1, e_1, e_1, f_1, f_1, f_1]\right\}$$

Ordering Symbols

```
𝕆[poly_, specs___] := Module[{vs, us, z},
  vs = Join @@ (First /@ {specs});
  us = Join @@ ({specs} /. (l_ → s_) :> (l /. x_i_ :> x_s));
  Simp@Total[CoefficientRules[Normal@Series[poly, {ħ, 0, $TD}], vs] /. (p_ → c_) :> c UU @@ (us^p)]
 ]
```

Theorem. $R = e^{h\otimes l + e\otimes f} = \mathbb{O}\left(\exp\left(hl + \frac{e^h - 1}{h} ef \,\middle|\, e\otimes lf\right)\right)$.

Brute Proof.

```
$TD = 6; 𝕆[Exp[ħ h_1 l_2 + (e^{ħ h_1} - 1)/h_1 e_1 f_2], {e_1} → 1, {l_2, f_2} → 2] == R_{1,2}
```

True

Debts.

1. How does the 𝕆 program work?

2. Why is the above theorem true?
3. What do you do with it?

old above / new below

Theorem. $R = e^{h\otimes l + e\otimes f} = \mathbb{O}\left(\exp\left(hl + \frac{e^h-1}{h}\,\mathrm{ef}\;\middle|\; e\otimes lf\right)\right)$.

Gentler Proof, Presented Brutely.

```
MatrixForm /@ {MatrixExp[h ρl + e ρf], MatrixExp[h ρl].MatrixExp[(ⅇ^h - 1)/h e ρf]}
```

$$\left\{\begin{pmatrix}1 & \frac{e\,(-1+e^h)}{h} & 0\\ 0 & e^h & 0\\ 0 & 0 & 1\end{pmatrix}, \begin{pmatrix}1 & \frac{e\,(-1+e^h)}{h} & 0\\ 0 & e^h & 0\\ 0 & 0 & 1\end{pmatrix}\right\}$$

The Invariant of the Trefoil, Again

```
Timing[$TD = 3;
 𝕆[Exp[ℏ h l1 + (ⅇ^(ℏ h) - 1)/h e4 f1 + ℏ h l5 + (ⅇ^(ℏ h) - 1)/h e2 f5 + ℏ h l3 + (ⅇ^(ℏ h) - 1)/h e6 f3],
   {l1, f1, e2, l3, f3, e4, l5, f5, e6} → 1] /. h1 → h]
```

$$\Big\{0.0625,\ \left(1 - 2h\hbar + h^2\hbar^2 + \frac{2h^3\hbar^3}{3}\right)U[\,] + \left(3h\hbar - 6h^2\hbar^2 + 3h^3\hbar^3\right)U[l_1] + \left(3\hbar - \frac{3h\hbar^2}{2} - \frac{3h^2\hbar^3}{2}\right)U[e_1, f_1] + \left(\frac{9h^2\hbar^2}{2} - 9h^3\hbar^3\right)U[l_1, l_1] + \left(9h\hbar^2 - \frac{9h^2\hbar^3}{2}\right)U[e_1, l_1, f_1] + \frac{9}{2}h^3\hbar^3 U[l_1, l_1, l_1] + \left(\frac{9\hbar^2}{2} + \frac{9h\hbar^3}{2}\right)U[e_1, e_1, f_1, f_1] + \frac{27}{2}h^2\hbar^3 U[e_1, l_1, l_1, f_1] + \frac{27}{2}h\hbar^3 U[e_1, e_1, l_1, f_1, f_1] + \frac{9}{2}\hbar^3 U[e_1, e_1, e_1, f_1, f_1, f_1]\Big\}$$

```
Timing[$TD = 5;
 𝕆[Exp[ℏ h l1 + (ⅇ^(ℏ h) - 1)/h e4 f1 + ℏ h l5 + (ⅇ^(ℏ h) - 1)/h e2 f5 + ℏ h l3 + (ⅇ^(ℏ h) - 1)/h e6 f3],
   {l1, f1, e2, l3, f3, e4, l5, f5, e6} → 1] /. h1 → h]
```

$$\Big\{1.09375,\ \left(1 - 2h\hbar + h^2\hbar^2 + \frac{2h^3\hbar^3}{3} - \frac{5h^4\hbar^4}{12} - \frac{23h^5\hbar^5}{30}\right)U[\,] + \left(3h\hbar - 6h^2\hbar^2 + 3h^3\hbar^3 + 2h^4\hbar^4 - \frac{5h^5\hbar^5}{4}\right)U[l_1] + \left(3\hbar - \frac{3h\hbar^2}{2} - \frac{3h^2\hbar^3}{2} + \frac{7h^3\hbar^4}{8} + \frac{61h^4\hbar^5}{40}\right)U[e_1, f_1] + \left(\frac{9h^2\hbar^2}{2} - 9h^3\hbar^3 + \frac{9h^4\hbar^4}{2} + 3h^5\hbar^5\right)U[l_1, l_1] + \left(9h\hbar^2 - \frac{9h^2\hbar^3}{2} - \frac{9h^3\hbar^4}{2} + \frac{21h^4\hbar^5}{8}\right)U[e_1, l_1, f_1] + \left(\frac{9h^3\hbar^3}{2} - 9h^4\hbar^4 + \frac{9h^5\hbar^5}{2}\right)U[l_1, l_1, l_1] + \left(\frac{9\hbar^2}{2} + \frac{9h\hbar^3}{2} + \frac{9h^2\hbar^4}{8} - \frac{3h^3\hbar^5}{8}\right)U[e_1, e_1, f_1, f_1] + \left(\frac{27h^2\hbar^3}{2} - \frac{27h^3\hbar^4}{4} - \frac{27h^4\hbar^5}{4}\right)U[e_1, l_1, l_1, f_1] + \left(\frac{27h^4\hbar^4}{8} - \frac{27h^5\hbar^5}{4}\right)U[l_1, l_1, l_1, l_1] + \left(\frac{27h\hbar^3}{2} + \frac{27h^2\hbar^4}{2} + \frac{27h^3\hbar^5}{8}\right)U[e_1, e_1, l_1, f_1, f_1] + \left(\frac{27h^3\hbar^4}{2} - \frac{27h^4\hbar^5}{4}\right)U[e_1, l_1, l_1, l_1, f_1] + \frac{81}{40}h^5\hbar^5 U[l_1, l_1, l_1, l_1, l_1] + \left(\frac{9\hbar^3}{2} + \frac{45h\hbar^4}{4} + \frac{117h^2\hbar^5}{8}\right)U[e_1, e_1, e_1, f_1, f_1, f_1] + \left(\frac{81h^2\hbar^4}{4} + \frac{81h^3\hbar^5}{4}\right)U[e_1, e_1, l_1, l_1, f_1, f_1] + \frac{81}{8}h^4\hbar^5 U[e_1, l_1, l_1, l_1, l_1, f_1] + \left(\frac{27h\hbar^4}{2} + \frac{135h^2\hbar^5}{4}\right)U[e_1, e_1, e_1, l_1, f_1, f_1, f_1] + \frac{81}{4}h^3\hbar^5 U[e_1, e_1, l_1, l_1, l_1, f_1, f_1] + \left(\frac{27\hbar^4}{8} + \frac{27h\hbar^5}{2}\right)U[e_1, e_1, e_1, e_1, f_1, f_1, f_1, f_1] + \frac{81}{4}h^2\hbar^5 U[e_1, e_1, e_1, l_1, l_1, f_1, f_1, f_1] + \frac{81}{8}h\hbar^5 U[e_1, e_1, e_1, e_1, l_1, f_1, f_1, f_1, f_1] + \frac{81}{40}\hbar^5 U[e_1, e_1, e_1, e_1, e_1, f_1, f_1, f_1, f_1, f_1]\Big\}$$

The Big $\mathfrak{g}_0$ Lemma.

1. $\mathbb{O}\left(e^{\gamma l + \beta e} \mid le\right) = \mathbb{O}\left(e^{\gamma l + e^{\gamma}\beta e} \mid el\right)$.
2. $\mathbb{O}\left(e^{\gamma l + \beta f} \mid fl\right) = \mathbb{O}\left(e^{\gamma l + e^{\gamma}\beta f} \mid lf\right)$.

3. $\mathbb{O}\left(e^{\beta e+\alpha f+\delta ef} \mid fe\right) = \mathbb{O}\left(\nu e^{\nu(-\alpha\beta h+\beta e+\alpha f+\delta ef)} \mid ef\right)$, with $\nu = (1+h\delta)^{-1}$.

Brute Proof.

```
$TD = 3; 𝕆[ℯ^(ħ γ l₁+β ħ e₁), {l₁, e₁} → 1]
```

$$U[] + \left(\beta\hbar + \beta\gamma\hbar^2 + \frac{1}{2}\beta\gamma^2\hbar^3\right)U[e_1] + \gamma\hbar U[l_1] + \left(\frac{\beta^2\hbar^2}{2} + \beta^2\gamma\hbar^3\right)U[e_1, e_1] + \left(\beta\gamma\hbar^2 + \beta\gamma^2\hbar^3\right)U[e_1, l_1] + \frac{1}{2}\gamma^2\hbar^2 U[l_1, l_1] + \frac{1}{6}\beta^3\hbar^3 U[e_1, e_1, e_1] + \frac{1}{2}\beta^2\gamma\hbar^3 U[e_1, e_1, l_1] + \frac{1}{2}\beta\gamma^2\hbar^3 U[e_1, l_1, l_1] + \frac{1}{6}\gamma^3\hbar^3 U[l_1, l_1, l_1]$$

```
$TD = 3; 𝕆[ℯ^(ħ γ l₁+ℯ^(n γ) β ħ e₁), {e₁, l₁} → 1]
```

$$U[] + \left(\beta\hbar + \beta\gamma\hbar^2 + \frac{1}{2}\beta\gamma^2\hbar^3\right)U[e_1] + \gamma\hbar U[l_1] + \left(\frac{\beta^2\hbar^2}{2} + \beta^2\gamma\hbar^3\right)U[e_1, e_1] + \left(\beta\gamma\hbar^2 + \beta\gamma^2\hbar^3\right)U[e_1, l_1] + \frac{1}{2}\gamma^2\hbar^2 U[l_1, l_1] + \frac{1}{6}\beta^3\hbar^3 U[e_1, e_1, e_1] + \frac{1}{2}\beta^2\gamma\hbar^3 U[e_1, e_1, l_1] + \frac{1}{2}\beta\gamma^2\hbar^3 U[e_1, l_1, l_1] + \frac{1}{6}\gamma^3\hbar^3 U[l_1, l_1, l_1]$$

```
$TD = 6; 𝕆[ℯ^(ħ γ l₁+β ħ e₁), {l₁, e₁} → 1] == 𝕆[ℯ^(ħ γ l₁+ℯ^(ħ γ) β ħ e₁), {e₁, l₁} → 1]
```

True

```
$TD = 6; 𝕆[ℯ^(ħ γ l₁+β ħ f₁), {f₁, l₁} → 1] == 𝕆[ℯ^(ħ γ l₁+ℯ^(n γ) β ħ f₁), {l₁, f₁} → 1]
```

True

```
$TD = 3; 𝕆[ℯ^(ħ (β e₁+α f₁+δ e₁ f₁)), {f₁, e₁} → 1]
```

$$\left(1 - \delta\hbar h_1 - \alpha\beta\hbar^2 h_1 + \delta^2\hbar^2 h_1^2 + 2\alpha\beta\delta\hbar^3 h_1^2 - \delta^3\hbar^3 h_1^3\right)U[] + \left(\beta\hbar - 2\beta\delta\hbar^2 h_1 - \alpha\beta^2\hbar^3 h_1 + 3\beta\delta^2\hbar^3 h_1^2\right)U[e_1] + \left(\alpha\hbar - 2\alpha\delta\hbar^2 h_1 - \alpha^2\beta\hbar^3 h_1 + 3\alpha\delta^2\hbar^3 h_1^2\right)U[f_1] + \left(\frac{\beta^2\hbar^2}{2} - \frac{3}{2}\beta^2\delta\hbar^3 h_1\right)U[e_1, e_1] + \left(\delta\hbar + \alpha\beta\hbar^2 - 2\delta^2\hbar^2 h_1 - 4\alpha\beta\delta\hbar^3 h_1 + 3\delta^3\hbar^3 h_1^2\right)U[e_1, f_1] + \left(\frac{\alpha^2\hbar^2}{2} - \frac{3}{2}\alpha^2\delta\hbar^3 h_1\right)U[f_1, f_1] + \frac{1}{6}\beta^3\hbar^3 U[e_1, e_1, e_1] + \left(\beta\delta\hbar^2 + \frac{1}{2}\alpha\beta^2\hbar^3 - 3\beta\delta^2\hbar^3 h_1\right)U[e_1, e_1, f_1] + \left(\alpha\delta\hbar^2 + \frac{1}{2}\alpha^2\beta\hbar^3 - 3\alpha\delta^2\hbar^3 h_1\right)U[e_1, f_1, f_1] + \frac{1}{6}\alpha^3\hbar^3 U[f_1, f_1, f_1] + \frac{1}{2}\beta^2\delta\hbar^3 U[e_1, e_1, e_1, f_1] + \left(\frac{\delta^2\hbar^2}{2} + \alpha\beta\delta\hbar^3 - \frac{3}{2}\delta^3\hbar^3 h_1\right)U[e_1, e_1, f_1, f_1] + \frac{1}{2}\alpha^2\delta\hbar^3 U[e_1, f_1, f_1, f_1] + \frac{1}{2}\beta\delta^2\hbar^3 U[e_1, e_1, e_1, f_1, f_1] + \frac{1}{2}\alpha\delta^2\hbar^3 U[e_1, e_1, f_1, f_1, f_1] + \frac{1}{6}\delta^3\hbar^3 U[e_1, e_1, e_1, f_1, f_1, f_1]$$

```
$TD = 6; With[{ν = (1 + ħ h δ)^-1},
 𝕆[ℯ^(ħ (β e₁+α f₁+δ e₁ f₁)), {f₁, e₁} → 1] == 𝕆[ν ℯ^(ħ ν (-ħ h α β+ β e₁+α f₁+δ e₁ f₁)), {e₁, f₁} → 1] /. h₁ → h
]
```

True

The Big g_0 Lemma. (again)

1. $\mathbb{O}\left(e^{\gamma l+\beta e} \mid le\right) = \mathbb{O}\left(e^{\gamma l+e^{\gamma}\beta e} \mid el\right)$.
2. $\mathbb{O}\left(e^{\gamma l+\beta f} \mid fl\right) = \mathbb{O}\left(e^{\gamma l+e^{\gamma}\beta f} \mid lf\right)$.
3. $\mathbb{O}\left(e^{\beta e+\alpha f+\delta ef} \mid fe\right) = \mathbb{O}\left(\nu e^{\nu(-\alpha\beta h+\beta e+\alpha f+\delta ef)} \mid ef\right)$, with $\nu = (1+h\delta)^{-1}$.

Gentler Proofs of 1 & 2 and of 3 at δ=0.

```
MatrixForm /@ {MatrixExp[γ ρl].MatrixExp[β ρe], MatrixExp[ℯ^γ β ρe].MatrixExp[γ ρl]}
```

$$\left\{\begin{pmatrix} 1 & 0 & 0 \\ 0 & e^{\gamma} & e^{\gamma}\beta \\ 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 1 & 0 & 0 \\ 0 & e^{\gamma} & e^{\gamma}\beta \\ 0 & 0 & 1 \end{pmatrix}\right\}$$

```
MatrixForm /@ {MatrixExp[β ρf].MatrixExp[γ ρl], MatrixExp[γ ρl].MatrixExp[ℯ^γ β ρf]}
```

$$\left\{\begin{pmatrix} 1 & e^{\gamma}\beta & 0 \\ 0 & e^{\gamma} & 0 \\ 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 1 & e^{\gamma}\beta & 0 \\ 0 & e^{\gamma} & 0 \\ 0 & 0 & 1 \end{pmatrix}\right\}$$

```
MatrixForm /@ {MatrixExp[α ρf].MatrixExp[β ρe], MatrixExp[-α β ρh].MatrixExp[β ρe].MatrixExp[α ρf]}
```

$$\left\{\begin{pmatrix}1 & \alpha & \alpha\beta \\ 0 & 1 & \beta \\ 0 & 0 & 1\end{pmatrix}, \begin{pmatrix}1 & \alpha & \alpha\beta \\ 0 & 1 & \beta \\ 0 & 0 & 1\end{pmatrix}\right\}$$

The Main $\mathfrak{g}_0$ Theorem.

Raw Version. The $\mathfrak{g}_0$ invariant of any S-component tangle T can be written in the form $Z(T) = \mathbb{O}\left(\omega e^{L+Q} \,\middle|\, \prod_{i\in S} e_i\, l_i\, f_i\right)$, where ω is a scalar (meaning, a rational function in the variables h_i and their exponentials $t_i = e^{h_i}$), where $L = \sum a_{ij}\, h_i\, l_j$ is a balanced quadratic in the variables h_i and l_j with integer coefficients a_{ij} and where $Q = \sum b_{ij}\, e_i\, f_j$ is a balanced quadratic in the variables e_i and f_j with scalar coefficients b_{ij}. Furthermore, after setting $h_i = h$ and $t_i = t$ for all i, the invariant $Z(T)$ is poly-time computable.

Proof. Indeed,

0. $R = e^{h\otimes l + e\otimes f} = \mathbb{O}\left(\exp\left(\text{hl} + \frac{e^h - 1}{h}\,\text{ef}\right) \,\middle|\, e\otimes \text{lf}\right)$,

1. $\mathbb{O}\left(e^{\gamma \text{l} + \beta \text{e}} \,\middle|\, \text{le}\right) = \mathbb{O}\left(e^{\gamma \text{l} + e^{\gamma}\beta \text{e}} \,\middle|\, \text{el}\right)$,

2. $\mathbb{O}\left(e^{\gamma \text{l} + \beta \text{f}} \,\middle|\, \text{fl}\right) = \mathbb{O}\left(e^{\gamma \text{l} + e^{\gamma}\beta \text{f}} \,\middle|\, \text{lf}\right)$,

3. $\mathbb{O}\left(e^{\beta \text{e} + \alpha \text{f} + \delta \text{ef}} \,\middle|\, \text{fe}\right) = \mathbb{O}\left(\nu e^{\nu(-\alpha\beta \text{h} + \beta \text{e} + \alpha \text{f} + \delta \text{ef})} \,\middle|\, \text{ef}\right)$, with $\nu = (1 + \text{h}\delta)^{-1}$,

and the rest is straight-forward.

Polished Version. With $\overline{e} = \frac{(e^h - 1)}{h} e$, the $\mathfrak{g}_0$ invariant of any S-component tangle T can be written in the form $Z(T) = \mathbb{O}\left(\omega^{-1} e^{L + \omega^{-1} Q} \,\middle|\, \prod_{i\in S} \overline{e}_i\, l_i\, f_i\right)$, where ω is a scalar (meaning, a polynomial in the variables $t_i = e^{h_i}$), where $L = \sum a_{ij}\, h_i\, l_j$ is a balanced quadratic in the variables h_i and l_j with integer coefficients a_{ij} and where $Q = \sum b_{ij}\, \overline{e}_i\, f_j$ is a balanced quadratic in the variables $\overline{e}_i$ and f_j with scalar coefficients b_{ij}. Furthermore, after setting $t_i = t$ for all i, the invariant $Z(T)$ is poly-time computable.

Debts.

1. Implement (and verify!).
2. Really prove part 3 of the big $\mathfrak{g}_0$-lemma.